<commit_before>@echo off

if NOT [%1] == [] (
    set msbuild=%1
    goto found
)

for %%v in (14.0, 12.0, 15.0, 4.0, 3.5, 2.0) do (
    for /F "usebackq tokens=2* skip=2" %%a in (
        `reg query "HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\MSBuild\ToolsVersions\%%v" /v MSBuildToolsPath 2^> nul`
    ) do if exist %%b (
        set msbuild="%%bmsbuild.exe"
        goto found
    )
)

set bat=%~nx0
echo MSBuild Tools was not found. Please define it manually like: `%bat:~0,-4% "full_path_to_msbuild.exe"` 1>&2

goto exit

:found

echo MSBuild Tools: %msbuild%

%msbuild% gnt.core /p:ngconfig="packages.config|RGiesecke.DllExport/packages.config|RGiesecke.DllExport.MSBuild/packages.config" /nologo /verbosity:m
%msbuild% "DllExport.sln" /verbosity:normal /l:"packages\vsSBE.CI.MSBuild\bin\CI.MSBuild.dll" /m:8 /t:Build /p:Configuration=Release

:exit<commit_msg>fixed CI build - forgotten packages.config<commit_after>@echo off

if NOT [%1] == [] (
    set msbuild=%1
    goto found
)

for %%v in (14.0, 12.0, 15.0, 4.0, 3.5, 2.0) do (
    for /F "usebackq tokens=2* skip=2" %%a in (
        `reg query "HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\MSBuild\ToolsVersions\%%v" /v MSBuildToolsPath 2^> nul`
    ) do if exist %%b (
        set msbuild="%%bmsbuild.exe"
        goto found
    )
)

set bat=%~nx0
echo MSBuild Tools was not found. Please define it manually like: `%bat:~0,-4% "full_path_to_msbuild.exe"` 1>&2

goto exit

:found

echo MSBuild Tools: %msbuild%

%msbuild% gnt.core /p:ngconfig="packages.config|NSBin/packages.config|RGiesecke.DllExport/packages.config|RGiesecke.DllExport.MSBuild/packages.config" /nologo /verbosity:m
%msbuild% "DllExport.sln" /verbosity:normal /l:"packages\vsSBE.CI.MSBuild\bin\CI.MSBuild.dll" /m:8 /t:Build /p:Configuration=Release

:exit
{'fraction_non_alphanumeric': 0.15492957746478872, 'fraction_numerical': 0.029295774647887324, 'mean_word_length': 3.8, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5843166', 'n_tokens_mistral': 740, 'n_tokens_neox': 704, 'n_words': 157}